

Prepayments

In Chapter 2, a notional amortization schedule was created that provided a basis for cash flowing into a transaction. This cash flow assumes that every loan in the pool will make payments exactly as they are scheduled. If it were that simple many finance professionals would be out of a job. In reality, cash flow becomes irregular when obligors choose to pay more than the scheduled amount, a prepayment, or when they do not pay at all, a delinquency, which sometimes becomes a default.

This chapter focuses on prepayments, how they are tracked, how they are projected, and how they are used in a cash flow model. First, it should be clear exactly what is considered a prepayment. A prepayment can either be “complete,” where the outstanding balance is paid off in full, or “partial,” where only a portion of the outstanding balance is paid. For example, an individual mortgage obligor might have extra money one month and decide to pay \$2,000 towards her mortgage when the scheduled payment is \$1,800. Since the periodic principal and interest is calculated in the \$1,800 due, the only use of the additional \$200 is to reduce the outstanding balance of the loan. This would be considered a partial prepayment. The other type of prepayment is where the entire balance is paid off. This can be the result of refinancing, credit-related events, and, on occasion, calculated due to foreclosure.

The latter part of this chapter features Model Builder exercises; but some of these exercises are not directly involved in building the example model. One exercise demonstrates how to take a file of raw historical prepayment information and create a projected prepayment curve. This projected prepayment curve then is used for prepayments in Project Model Builder. The purpose of this is to show a reader three important parts of structured finance modeling:

1. The ideal format for prepayment data.
2. How to convert the prepayment data into projections.
3. How to apply and integrate the projections in a working financial model.

HOW PREPAYMENTS ARE TRACKED

While the concept of a prepayment is relatively simple, there are numerous methods of tracking prepayments and calculating prepayment rates. Prepayment terminology

cannot only differ between asset classes, but even between companies within the same industry. It is important to understand what each prepayment expression means and the underlying calculation involved so historical data can be interpreted and used to make projections. Prior to exploring how to work with prepayment data, it would be useful to define some of the primary terms.

SMM: Single Monthly Mortality

The most basic measurement of prepayment is known as *single monthly mortality* (SMM). The name sounds a bit morbid, but it comes from the idea that the prepayment is actually getting rid of or killing off the asset. In most modeling, whether it is a mortgage or auto model, the prepayment rate is typically converted to SMM for calculating prepayments. This is done because most models are done on a monthly basis and as the name implies SMM is a monthly calculation. It is calculated by taking the dollar amount of the prepayment over the previous month's outstanding balance (less the scheduled principal payment). The formula for SMM is:

$$SMM = \frac{\text{Dollar amount of period } n \text{ prepayment}}{\text{Beginning of period } n \text{ current balance} - \text{Period } n \text{ scheduled principal payment}}$$

CPR: Conditional Prepayment Rate

Since most transactions are monthly, SMM is appropriate for calculations. However, for mortgage models and other long-term assets the prepayment rate is often expressed as an annual rate known as the *conditional prepayment rate* (CPR). CPR is an often misunderstood term with incorrect acronym explanations such as Cumulative Prepayment Rate or confusion between CDR, which is a default term that will be described in Chapter 4. The most basic definition of CPR is that it is an annualized calculation of SMM. To convert SMM to CPR use the following formula:

$$CPR = 1 - (1 - SMM)^{12}$$

This calculation can be set up in a spreadsheet or is referenced in numerous books in a table format (i.e., SMM to CPR conversion tables).

PSA: Public Securities Association

With many years of historical prepayment data on mortgages, standardized prepayment amounts have been developed by the *Public Securities Association* (PSA). These curves or multiples of these curves are often used to model mortgage transactions. One-hundred percent PSA assumes prepayment rates of 0.2 percent CPR in the first month following origination of the mortgage loans (not the pool) and an additional 0.2 percent CPR in each succeeding month until the 30th month. In the




FIGURE 3.1 100 percent PSA expressed as a line graph.

30th month and beyond, 100 percent PSA assumes a fixed annual prepayment rate of 6.0 percent CPR. See Figure 3.1 for a graph of the standard PSA curve.

To calculate the prepayment rate for any specific multiple of PSA, adjust the annual rate at 100 percent PSA by that multiple. For example, 200 percent PSA assumes prepayment rates equal to twice the CPRs from the 100 percent PSA model, on a pool-by-pool basis.¹ Specifically, 0.2 percent CPR would instead be 0.4 percent CPR for the first month, following the origination of the mortgage loans, and grow by 0.4 percent CPR for each of the successive 29 months and 12.0 percent CPR for all of the remaining months.

ABS: Absolute Prepayment Speed

For shorter term assets such as auto loans, a more appropriate measure of prepayment is the *absolute prepayment speed* (ABS). Due to the short terms of the asset, larger amounts of principal are paid faster causing an increasing prepayment rate as the pool seasons. To use ABS in a model it should be converted to SMM. This is done by the following formula:

$$ABS = 100 * SMM / (100 + SMM * (n - 1))$$

In this formula n is the number of periods that have transpired since origination. For instance, if a pool is known to have an ABS of 1.50 percent, the corresponding SMM to use in a calculation five months out would be 1.60 percent.

¹Public Securities Association, *Standard Formulas for the Analysis of Mortgage-Backed Securities and Other Related Securities*, 6/01/90, p. SF-5.

HISTORICAL PREPAYMENT DATA FORMATS

Projecting prepayments can get very complex with interest rate and economic expectations. The most basic starting point is to analyze how the assets have prepaid in the past. Obtaining sufficient historical prepayment data (at least three years worth) on the specific assets that are to be analyzed is the best approach to take. Depending on the company, this data will come in a variety of formats, but must include at minimum: a historical account of the asset's principal balance each period and the amount of principal that was prepaid each period. Ideally, though, the company provides a vintage analysis of the prepayments.

As the name implies, a *vintage analysis* tracks information by the month, quarter, or year that the assets were originated. For example, assets originated in August 1998 would have an original pool balance from that month. From August to September, there are prepayments for just the loans originated in August 1998, which should be tracked separately from the entire book of assets. Similarly, the current principal balance of the August 1998 vintage should be tracked separately. This type of tracking should be done for every month of origination because it allows for a proper trend analysis.

It should be noted that the data presented in the Model Builder section is an ideal example of how prepayment data is tracked. Often times this data is part of a larger analysis and may require manipulation to bring into a usable format.

BUILDING PREPAYMENT CURVES

With suitable prepayment data the next step is to build prepayment curves that represent each vintage and an aggregated curve that represents the assets as a whole. The first point to remember is that prepayment amounts used in Project Model Builder will be calculated off of SMM, which is the prepayment per period over the current balance of the period. If this is not done, then the rates will have to be calculated. Once they are calculated there should be a table of rates that have origination periods on one axis and periods on the other axis. This becomes clearer as the Model Builder exercise for this chapter is worked. An example for you to reference now is shown in Figure 3.2.

The table of monthly SMM rates can then be aggregated by each period using a weighted average methodology. To account for differences in balances, each SMM should be weighted by the periodic principal balance from the vintage that the SMM is derived from. The weighted average aggregate curve can then be summed to create a cumulative curve. Be aware that this is the most fundamental method for creating prepayment curves and more advanced modeling would include an array of different factors and sophisticated techniques.

As an introduction the weighted average curve is the most basic representation of the assets prepayment behavior. Using this curve to project prepayments assumes the following:

	A	B	C	D	E	F	G	H	I	J	K	L
1	Prepayment Analysis											
2												
3		Monthly SMM										
4												
5		Originations										
6		Jan-04	Feb-04	Mar-04	Apr-04	May-04	Jun-04	Jul-04	Aug-04	Sep-04	Oct-04	
7	Periods Out	0										
8		1	0.78%	0.90%	0.84%	0.71%	1.01%	1.55%	0.49%	0.87%	0.41%	0.62%
9		2	0.02%	0.51%	1.47%	0.93%	0.30%	1.17%	0.26%	0.15%	0.44%	0.76%
10		3	1.55%	1.38%	0.16%	1.06%	0.53%	0.97%	1.49%	0.88%	0.57%	0.97%
11		4	0.41%	0.40%	0.37%	1.09%	0.13%	0.96%	0.23%	0.73%	0.22%	0.76%
12		5	0.16%	1.27%	0.77%	0.43%	0.44%	0.98%	1.49%	0.19%	0.44%	1.29%
13		6	0.10%	0.51%	1.37%	0.82%	1.43%	0.39%	1.28%	1.12%	0.54%	0.93%
14		7	1.05%	1.04%	0.71%	0.26%	1.19%	0.35%	0.63%	1.36%	0.47%	0.84%
15		8	0.57%	1.37%	1.37%	0.54%	1.12%	0.96%	1.30%	1.25%	0.38%	0.43%
16		9	0.62%	1.07%	0.40%	1.17%	0.14%	1.40%	0.80%	0.45%	1.26%	1.65%
17		10	1.06%	0.40%	1.12%	0.84%	1.13%	0.17%	1.16%	0.73%	1.35%	0.34%

FIGURE 3.2 A standard analysis of prepayment rates is completed by tracking each origination period separately over time.

1. That the assets requiring prepayment projections are the same as the assets used to create the projected prepayment curve. This might sound like simple logic, but if the asset class is for example mortgages and the historical prepayment data is generated off of fixed rate mortgages, the prepayment curve calculated should only be used for fixed rate mortgages from that company. This distinction should even be as granular as the precise type of mortgage, since prepayment rates for mortgages are very particular to the type of product.
2. That there is no significant trend in the prepayment rates. If a noticeable increasing or decreasing trend is apparent in the historical prepayment rates and this trend is expected to maintain, it should be accounted for in the projected curve. Sometimes a shift in market focus, origination process, or economic environment causes such a trend, which could affect future prepayment performance.
3. While more complex than this text will demonstrate in the examples, a consideration for any prepayment curve should be interest rates. Prepayment behavior is closely linked to interest rates. A simple example is a homeowner with a mortgage, which could be fixed or floating. If the mortgage is fixed rate and it was originated during a high interest rate environment, the homeowner will be more inclined to refinance when interest rates decline. Or, if a floating rate mortgage was originated during a low interest rate environment, the obligor would want to refinance to a fixed rate if he or she expects interest rates to increase. In more advanced prepayment analyses, expected interest rate scenarios should be run using a variety of prepayment curves.

PREPAYMENT CURVES IN PROJECT MODEL BUILDER

Once the projected curve is created it will be stored in the model. For maximum flexibility the model should have room for multiple custom curves, conversion

formulas for the main types of prepayment rates (as described above), and standard prepayment curves. Also, the model should be developed in a way that it is easy to select which curve to use, the option to use different curves for specific groups of assets, and a method of automating scenarios so curves can be selected and run without much adjustment. All of this will be demonstrated and worked through in the Model Builder exercises.

THE EFFECT OF PREPAYMENTS ON STRUCTURED TRANSACTIONS

Prior to the step-by-step exercises in the Model Builder sections a moment should be taken to understand the effects that prepayments have in a structured finance transaction. Prepayments accelerate the repayment of principal for the assets and, in pass-through structures, the principal of the notes or certificates. At first thought an investor might think prepayments are good since they will have their principal returned to them faster.

This is true if the assets are consistently performing poorly and there is risk that there will not be enough cash flow to repay the investor. In such a case, one would want their money as fast as possible. However, when assets are performing as they are expected, there should be more asset interest than *liability cost*—also known as *excess spread*. This excess spread provides credit protection over the life of a transaction. Obligor who tend to prepay the fastest are the better credits with high interest rates that, when, eliminated, reduce excess spread. If those higher interest obligors prepay quickly, the transaction terminates faster and overall there will be less excess spread in the deal than if the assets prepaid at a slower rate. This is extremely important because if a comparison were made between two transactions—A and B—where everything was the same including default rate but the only difference was transaction A prepaid faster than transaction B, then transaction A would have less excess spread.

Another consideration is that the prepayment rate needs to be modeled accurately to give a correct estimation of the weighted average life of the transaction and the total return provided to investors. Institutional investors often purchase asset-backed securities with certain tenors and returns in mind. Prepayments can drastically affect the weighted average life of a transaction and can reduce the overall interest paid, thereby reducing the total return of investment.

MODEL BUILDER 3.1: HISTORICAL PREPAYMENT ANALYSIS AND CREATING A PROJECTED PREPAYMENT CURVE

1. The first step in any prepayment analysis is to review the data that exists. Open *MB3-1_Raw_Data.xls* from the Ch03 folder on the CD-ROM. There is only one sheet in this workbook and it contains very pertinent and organized data for

a prepayment analysis. The data included on the sheet is the minimum amount necessary to complete a prepayment analysis: pool balances and prepayment amounts.

2. To get started with the analysis, create another sheet and name it **Prepay Analysis**. This sheet is where the SMM prepayment rates will be calculated and aggregated by pool. Since the data will be organized similar to the raw data, copy cells A5:AA31 from the Raw Data sheet and paste them on A5 of the Prepay Analysis sheet.
3. Clear the contents of cells A7:AA7 because there will be no prepay rates in time 0. The formula to calculate the SMM rates is the prepay amount for the current period divided by the beginning of period balance. In the official formula, scheduled principal should be deducted from the beginning of period balance; however, in this case there is no information on how the balance is reduced each month (that is, what portion is scheduled principal versus default). The difference should not have a significant impact so using the beginning of period balance is sufficient. The Excel formula to enter in cell C8 is:

$$='Raw Data'!C39/'Raw Data'!C7$$

4. When this is copied from C8:AA31, there is a minor problem with #DIV/0 errors. This can be easily taken care of with an IF statement. Modify cell C8 as follows:

$$=IF('Raw Data'!C39="", "", 'Raw Data'!C39/'Raw Data'!C7)$$

This will prevent any 0 balances from trying to calculate and stop division by zero errors from populating across the sheet. At this point, the analysis should look like an upside down triangle as pictured in Figure 3.3.

5. With each vintage's prepayment rates laid out over time, the next step is to create an aggregate curve that represents how the asset's prepay on average. Since the vintages have different balances a weighted average should be used for the rates. The easiest way to calculate a weighted average in Excel is using a SUMPRODUCT-SUM combination. If this has never been done before, there is a detailed explanation in the Toolbox section of this chapter. There is a slight difference when using this combination for curves because there could be zero values that could skew the average. To eliminate this potential problem, a count of relevant cells needs to be created.
6. Label cell AC6 **WA Count**. The cells in AC8 through AC31 will contain a numeric value that represents the relevant number of cells of data for each period that should be calculated in the weighted average. For example, row 8 contains data for the first period of each vintage, row 9 the data for the second vintage, and so on. Notice though that the data is a triangle and the further the periods out, the less data exists. This is logical because if the current month is the beginning of January 2006 there should be no data for January 2006 one month out nor two months of data for loans originated in December 2005. To

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
1	Prepayment Analysis																
2																	
3		Monthly SMM															
4																	
5			Originations														
6			Jan-04	Feb-04	Mar-04	Apr-04	May-04	Jun-04	Jul-04	Aug-04	Sep-04	Oct-04	Nov-04	Dec-04	Jan-05	Feb-05	Mar-05
7	Periods Out	0															
8		1	0.78%	0.80%	0.84%	0.71%	1.01%	1.55%	0.45%	0.87%	0.41%	0.62%	1.2%	0.62%	0.12%	1.61%	0.88%
9		2	0.02%	0.71%	1.47%	0.70%	0.70%	1.47%	1.27%	0.10%	0.44%	0.70%	0.00%	0.4%	1.77%	0.00%	
10		3	1.56%	1.38%	0.18%	1.18%	1.65%	0.94%	1.4%	0.88%	0.57%	0.94%	0.10%	0.89%	0.22%	1.4%	0.75%
11		4	0.41%	0.40%	0.37%	1.29%	1.5%	0.88%	0.22%	0.70%	0.02%	0.78%	0.54%	0.44%	0.44%	1.45%	1.50%
12		5	0.18%	1.41%	0.74%	0.49%	1.44%	0.95%	1.4%	0.19%	0.44%	1.20%	0.44%	0.18%	0.1%	1.2%	0.94%
13		6	0.10%	0.51%	1.37%	0.22%	1.65%	0.38%	1.22%	1.12%	0.54%	0.89%	1.27%	0.12%	0.4%	1.81%	0.75%
14		7	0.05%	1.94%	0.74%	0.70%	1.15%	0.17%	0.6%	1.30%	0.47%	0.84%	0.73%	0.00%	1.4%	1.0%	1.21%
15		8	0.57%	1.37%	1.37%	0.54%	1.12%	0.96%	1.37%	1.25%	0.68%	0.43%	0.90%	0.84%	1.47%	1.12%	1.12%
16		9	0.62%	1.07%	0.40%	1.17%	0.14%	1.40%	0.0%	0.46%	1.20%	1.55%	1.5%	1.28%	1.2%	1.80%	1.18%
17		10	1.08%	0.40%	1.12%	0.54%	1.15%	0.17%	1.1%	0.73%	1.08%	0.34%	0.84%	0.88%	1.02%	0.37%	
18		11	0.38%	0.47%	0.47%	0.20%	0.25%	0.04%	0.4%	0.51%	1.28%	0.75%	0.19%	1.89%	0.1%	1.17%	
19		12	1.80%	1.44%	0.36%	0.25%	0.25%	0.23%	0.02%	0.71%	0.73%	0.42%	1.40%	1.33%	0.7%		
20		13	0.34%	0.57%	0.74%	0.18%	0.17%	0.00%	1.5%	1.10%	0.60%	1.4%	0.77%				
21		14	1.10%	1.12%	0.28%	1.21%	0.55%	0.51%	1.1%	0.90%	0.91%	0.48%	0.38%				
22		15	0.10%	1.47%	1.54%	1.14%	1.05%	1.10%	1.74%	0.00%	1.1%	0.00%					
23		16	1.3%	0.25%	0.65%	0.51%	1.12%	1.11%	0.51%	1.17%	1.20%						
24		17	1.56%	0.93%	0.81%	0.27%	0.05%	1.19%	0.5%	1.1%							
25		18	0.83%	0.51%	1.15%	0.14%	1.81%	1.52%	0.2%								
26		19	1.51%	1.44%	1.27%	1.22%	1.25%	1.37%									
27		20	0.21%	0.51%	1.13%	1.12%	0.97%										
28		21	1.53%	0.56%	1.03%	1.46%											
29		22	0.54%	1.36%	1.36%												
30		23	0.43%	0.51%													
31		24	1.53%														

FIGURE 3.3 The raw prepayment data should be converted into percents and cleaned up at this point.

help with the SUMPRODUCT-SUM combination a numeric count of the data that should be included in the calculation is necessary. This can be accomplished by using the COUNT formula in conjunction with data specific modifications.

- In cell AC7, use COUNT on the cells C6 through AA6. This produces a value of 25. Lock this reference down. For the first period in cell AC7, there is one period of unknown data: January 2006. In the next period there will be two periods of unknown data: January 2006 and December 2005. To use a single formula to not count those months, subtract the current period from the count. Since the numeric periods exist in column B, use those as the reference. The final formula should look like:

$$= \text{COUNT}(\$C\$6:\$AA\$6) - B8$$

Copy and paste this formula into cells AC8 to AC31. The result should be a vertical row of numbers that decrease as the periods decrease as shown in Figure 3.4.

- Next label AD6 WA SMM Curve. This is where the weighted averages (WA) are calculated. Any weighted average of A data is the sum of the products of A data and B data divided by the sum of B data. This can be accomplished in Excel using SUMPRODUCT and SUM. An additional element of complexity is making sure not to average blank cells. This is done using the OFFSET function within the other formulas. In cell AD8 enter the following formula:

$$= \text{SUMPRODUCT}(C8:\text{OFFSET}(\$B8,0,\$AC8),\text{'Raw Data'!C7}:\text{OFFSET}(\text{'Raw Data'!B7,0,\$AC8}))/\text{SUM}(\text{'Raw Data'!C7}:\text{OFFSET}(\text{'Raw Data'!B7,0,\$AC8}))$$

	AB	AC	AD	AE
1				
2				
3				
4				
5				
6		WA Count	WA SMM Curve	
7				
8		24	0.80%	
9		23	0.86%	
10		22	0.73%	
11		21	0.68%	
12		20	0.73%	
13		19	0.80%	
14		18	0.68%	
15		17	1.08%	
16		16	0.94%	
17		15	0.64%	
18		14	0.50%	
19		13	0.75%	
20		12	0.77%	
21		11	0.82%	
22		10	0.97%	
23		9	0.97%	
24		8	0.84%	
25		7	0.85%	
26		6	1.31%	
27		5	0.90%	
28		4	1.20%	
29		3	1.03%	
30		2	0.51%	
31		1	1.53%	

FIGURE 3.4 Notice the count function and the WA SMM Curve.

The SUMPRODUCT is taking two rows of data: the monthly SMMs (row 8 of the Prepay Analysis sheet) and the beginning of period balances for each vintage (row 7 of the Raw Data sheet). Instead of taking the entire row of data though, the SUMPRODUCT reference uses an OFFSET function to instruct the SUMPRODUCT to only take a certain amount of cells. This OFFSET uses the count system of relevant data created in column AC. This way the SUMPRODUCT only references the relevant data. Similarly, the SUM of the balances that is used as the divisor only takes in the relevant balances. If this OFFSET didn't exist the SUM would be much higher than necessary. Copy and paste the formula into cells AD8:AD31.

9. Create an additional sheet and name it **Summary**. This is where the curves that will be used for the projected modeling will be stored. Copy cells A7:B31 from the Prepay Analysis sheet and paste them in A7 of the Summary sheet. On the Summary sheet, label D5 **WA SMM Curve**. In D8 through D31, reference the

WA SMM curve from the Prepay Analysis sheet. Typically this is what will be used in a model to forecast prepayments. It has been simplified with only two years of data to keep the example simple, but in models like Project Model Builder a longer curve or estimation is typically necessary. This can be achieved by using more data, using a standard curve like PSA, or using rating agency assumptions that are often given in terms of CPR.

10. To see how SMM translates into CPR and vice versa, label cell E5 **CPR**. In cell E8, use the following formula to convert to CPR:

$$=1-(1-D8)^{12}$$

Copy and paste this formula over the range E8:E31. These are the annualized rates of the SMMs. Rating agencies often give CPR assumptions for assets, so it is important to understand how to go back and forth between calculations. The final prepayment curve should look like Figure 3.5.

	A	B	C	D	E
1	Prepay Summary				
2					
3					
4					
5				WA SMM Curve	CPR
6					
7	Periods Out	0			
8		1		0.80%	9.17%
9		2		0.86%	9.84%
10		3		0.73%	8.39%
11		4		0.68%	7.87%
12		5		0.73%	8.47%
13		6		0.80%	9.19%
14		7		0.68%	7.83%
15		8		1.08%	12.19%
16		9		0.94%	10.75%
17		10		0.64%	7.47%
18		11		0.50%	5.85%
19		12		0.75%	8.65%
20		13		0.77%	8.81%
21		14		0.82%	9.45%
22		15		0.97%	10.99%
23		16		0.97%	11.01%
24		17		0.84%	9.65%
25		18		0.85%	9.72%
26		19		1.31%	14.59%
27		20		0.90%	10.23%
28		21		1.20%	13.51%
29		22		1.03%	11.64%
30		23		0.51%	5.95%
31		24		1.53%	16.88%

FIGURE 3.5 The final prepayment curve is expressed in SMM and CPR.

MODEL BUILDER 3.2: INTEGRATING PROJECTED PREPAYMENTS IN ASSET AMORTIZATION

1. The last part that was completed in Project Model Builder was the creation of a notional amortization schedule. This Model Builder section will begin creating the actual amortization schedule beginning with prepayments. Go to the Cash Flow sheet in Project Model Builder and label cell L3 **Actual Amortization**. Below this, in row 4, is a series of column headers. Label the following cells as described:

L4: **Beginning Balance**

M4: **Default Rate**

N4: **New Defaults**

O4: **Amort Factor**

P4: **Prepay Rate**

Q4: **Voluntary Prepay**

R4: **Actual Amort**

S4: **Interest Rate**

T4: **Actual Interest**

U4: **Principal Recovery**

V4: **Ending Balance**

Many of these columns are placeholders for chapters to come; but they should be created now rather than inserting columns later.

2. Two of these columns contain formulas in period 0 (row 6): the Amort Factor and the Ending Balance, which as mentioned earlier is the starting balance when it is in time period 0. The Amort Factor is the ratio of the current notional balance to the original notional balance. This is important because it is an indication of how the asset balance is reduced on a scheduled basis and will be used to help determine prepay amounts. In period 0, however, it is always 1 since no amortization has taken place. Otherwise it is the end of period scheduled balance over the original scheduled balance. This is the formula that should be entered in O6 and copied and pasted through O366:

$$=IF(A6=0,1,J6/\$J\$6)$$

The Ending Balance will always be the starting asset current balance for period 0, but will reduce by scheduled amortization, prepayments, and defaults. Even though the cells are empty right now use the following references in the formula for cell V6:

$$=IF(A6=0,AssetCurBal1,L6-N6-Q6-R6)$$

Copy and paste this formula into the range V6:V366.

3. The next column to work on is column L, the Beginning Balance. As in the notional schedule, the beginning balance will always be the ending balance from

the period prior. In cell L7, enter the reference =V6 and copy and paste it down through the range L6:L366.

4. Before proceeding to the Prepay Rate, the Inputs sheet needs to be updated so Prepayment Inputs can be changed quickly. On the Inputs sheet, merge cells B16:O16 and use the long cell as a label that states **PREPAYMENT/DEFAULTS/RECOVERY INPUTS**. Also add the following labels:
 B17: **Description**
 C17: **Prepay Curve**
 D17: **Prepay Stress**
5. The Description is a reference to the description of the assets that the prepayment, default, and recovery information will affect. This is Asset Pool 1, which was named **AssetDes1**. In B18, enter =**AssetDes1**. Name this cell **pdrDes1**.
6. C18 is where the prepayment curve will be selected from the Vectors sheet. This is very similar to the asset interest rate curve selection. Like that one there needs to be a selection of possible curves on the Vectors sheet. Go to the Vectors sheet and enter the following labels into these cells:
 M5: **SMM 1**
 N5: **SMM 2**
 O5: **SMM 3**
 P5: **Custom CPR 1 DATA ENTRY ONLY**
 Q5: **Custom CPR 2 DATA ENTRY ONLY**
 R5: **Custom CPR 3 DATA ENTRY ONLY**
 Select and name the range M5:O5 **lstPrepayCurve**.
7. Go back to the Inputs sheet and select C18. Using data validation, make the possible inputs for this cell the range **lstPrepayCurve**. There should only be three possible inputs because the range only included the SMM ranges. The CPR columns created on the vector sheet, which are labeled **DATA ENTRY ONLY**, are used to put CPR curves in, but must be converted to SMM for use in the model. Name C18 **pdrPrepay1**.
8. D18 contains the stressor for prepayments. This is a numeric multiple that is multiplied against the prepayment curve at each period. For now enter a 1 in cell D18 and name the cell **pdrPrepayStress1**.
9. To follow in the example calculations, copy and paste the existing prepayment curve, which is present in the completed section of the Vectors sheet in this *MB3-2.xls* in the Ch03 folder on the CD-ROM. This curve should be pasted in the same place, cells M7:M366. (In the Ch03 folder on the CD-ROM, there is an example in the Additional Files subfolder that contains the PSA curve calculation.)
10. The next step is having the correct prepayment rate show up on the Cash Flow sheet. This is accomplished in a similar manner to the asset interest rate, using an **OFFSET-MATCH** combination, but is much simpler since there are no caps, floors, or rate resets. The numerical value that should show up in P7 depends on the prepay curve selected on the Inputs sheet and the period (in this

case 1). Since the values are stored on the Vectors sheet, the OFFSET will start there with Vectors L5. To get the correct rate Vectors L5 needs to be offset by the period that the formula is in (1 in the case of Cash Flow P7) and the prepay curve (the selected one in this example is SMM 1, which is the first curve in the list). An additional factor that needs to be taken into consideration is the seasoning. Seasoned assets need to reference their correct timing on the prepayment curve, otherwise the wrong historically created prepayment percents will be applied to the asset balances. The following formula will accomplish this and change as different curves are selected on the input sheet and as the periods change:

$$= \text{OFFSET}(\text{Vectors!}\$L\$6, A7 + \text{Age}1, \text{MATCH}(\text{pdrPrepay}1, \text{lstPrepayCurve}, 0)) * \text{pdrPrepayStress}1$$

Copy and paste this formula from P7 through P366.

11. The final formula of this Model Builder exercise is the most critical. It determines the actual dollar amount of prepayments for the period. The first part of this formula is an IF function for clean up purposes. If there is a zero balance and the prepay calculation is attempted then there will be #DIV/0 errors. To prevent this start the formula in Q7 with:

$$= \text{IF}(L7=0, 0,$$

The next parts of the formula are MAX and MIN formulas. The MAX formula ensures that in cases of high default there is no negative balance that could skew prepayments. The MIN formula takes the lesser of the balance less defaults and the calculated prepay amount. The balance less defaults is the Beginning Balance minus New Defaults (L7-N7). The prepayment calculation is the Beginning Balance multiplied by the percentage asset reduction in the notional schedule multiplied by the monthly prepayment rate (as measured in SMM). A confusing part of the last statement is what “the percentage asset reduction in the notional schedule” is and why it is part of the calculation. This is the current amort factor divided by the last month’s amort factor. This represents the scheduled amortization on a fractional basis and removes it from the prepayment amount. The final formula should look like:

$$= \text{IF}(L7=0, 0, \text{MAX}(\text{MIN}((L7-N7), L7 * O7 / O6 * P7), 0))$$

In particular, notice that defaults (N7) are removed from the Beginning Balance as part of the MIN function. This is important because as loans approach their final periods they will have to be cleaned up. Defaults take precedence in calculations since it is assumed that a defaulted loan will not voluntarily prepay.

TOOLBOX

Weighted Averages Using SUMPRODUCT and SUM

When doing any analysis in finance on a pool of assets the word “average” comes up often: What is the average interest rate? How long is the average remaining term? What is the average age of the loans? The answers to all of these questions must be provided in terms of the *weighted average*, otherwise an incorrect estimation will most likely be made. The reason for this is that financial assets that are pooled together in a transaction rarely have homogenous principal balances from the start, with differences being more exacerbated as time progresses.

Principal balances are important because in a pool of assets, those with larger principal balances will have more influence on the characteristics and performance of the pool than assets with smaller balances. An easy example is to take a look at a pool of two loans, their balances, and their remaining terms. In the examples, the loans are quite different. One is fairly new and has a high principal balance and a long remaining term while the other is very seasoned and has a low principal balance and short remaining term. As seen in Figure 3.6 (and in Ch03’s Additional Files subfolder for the Model Builder files on the CD-ROM), the average remaining term and rate of this pool is taken by using the AVERAGE function in Microsoft Excel. This produces a result of 187.50 periods and a rate of 6.00 percent.

This is a misleading result because most of the pool’s money will be outstanding for much longer than 187.50 periods and at a far higher rate than 6.00 percent. The correct way to describe the average remaining term is to weight the average by the current principal balance. By using a weighted average based on current balance, the remaining term would be nearly 360 periods with a rate of 10 percent. This is done by multiplying the terms by the balances, summing up those products, and then dividing that sum by the sum of the principal. The weighted average can be calculated without using functions, but the calculation is made much easier by using SUMPRODUCT and SUM.

	A	B	C	D	E
1	Weighted Average Example				
2					
3			Balance	Rate	Term
4		Loan 1	1,000,000	10.00%	360
5		Loan 2	500	2.00%	15
6					
7		Average		6.00%	187.50
8		Weighted Average		10.00%	359.83

FIGURE 3.6 There is a noticeable difference between taking an arithmetic average and a weighted average.

SUMPRODUCT takes two arrays of equal length and multiplies each value in one array by the corresponding value of the second array, and then sums up all of those products. In Figure 3.6, for example, the remaining term array (E4:E5) is multiplied by the corresponding balances (C4:C5). SUMPRODUCT finishes by calculating the sum of all of these products. The arrays should be referenced in this manner:

=SUMPRODUCT(Array Reference 1, Array Reference 2)

SUMPRODUCT takes care of the first two parts of calculating a weighted average, so all that is left is to divide by the sum of the balances using the SUM function. This function combination can be used for calculating weighted averages for any loan characteristic and typically uses the current principal balance as the weighting.

Also, as seen in this chapter, the weighted average should be used for summarizing historical rates. Often, for such purpose, the OFFSET function needs to be used on the arrays to prevent zeros from skewing the average.